unittest
{
	assert( [ "3", "2 3 5", "3 4 1" ].parse.expand.solve == 3 );
	assert( [ "3", "2 3 3", "2 2 1" ].parse.expand.solve == 0 );
	assert( [ "3", "17 7 1", "25 6 14" ].parse.expand.solve == -1 );
	assert( [
		"12",
		"757232153 372327760 440075441 195848680 354974235 458054863 463477172 740174259 615762794 632963102 529866931 64991604",
		"74164189 98239366 465611891 362739947 147060907 118867039 63189252 78303147 501410831 110823640 122948912 572905212"
	].parse.expand.solve == 5 );
}



import std.algorithm;
import std.conv;
import std.range;
import std.stdio;
import std.string;
import std.typecons;

void main()
{
	stdin.byLineCopy.parse.expand.solve.writeln;
}

auto parse( Range )( Range input )
if( isInputRange!Range && is( ElementType!Range == string ) )
{
	auto n = input.front.to!( long );
	input.popFront;
	auto as = input.front.split.to!( long[] );
	input.popFront;
	auto bs = input.front.split.to!( long[] );
	return tuple( as, bs );
}

auto solve( long[] as, long[] bs )
{
	// 準備度の差を計算
	auto ds = new long[ as.length ];
	auto dt = 0L;
	foreach( ref d, a, b; lockstep( ds, as, bs ) )
	{
		d = a - b;
		dt += d;
	}
	
	// 差の合計がマイナスだったら絶対不合格
	if( dt < 0 ) return -1;
	
	// 差がマイナスの試験を集計
	auto ms = ds.filter!( ( d ) => d < 0 );
	auto mc = ms.walkLength;
	auto mt = -ms.sum;
	
	// 差がマイナスの試験が無かったら元から合格
	if( mc == 0 ) return 0;
	
	// 差がプラスの試験から差がマイナスの試験へ余分な準備度を移す（準備度が大きい順に使う）
	ds.sort!( ( a, b ) => b < a );
	auto ps = ds.filter!( ( d ) => 0 < d );
	auto pc = 0L;
	auto pt = 0L;
	foreach( p; ps )
	{
		if( mt <= pt ) break;
		pc++;
		pt += p;
	}
	
	return mc + pc;
}
